// File: @chainlink/contracts/src/v0.8/interfaces/AggregatorV3Interface.sol


pragma solidity ^0.8.0;

interface AggregatorV3Interface {
  function decimals() external view returns (uint8);

  function description() external view returns (string memory);

  function version() external view returns (uint256);

  function getRoundData(
    uint80 _roundId
  ) external view returns (uint80 roundId, int256 answer, uint256 startedAt, uint256 updatedAt, uint80 answeredInRound);

  function latestRoundData()
    external
    view
    returns (uint80 roundId, int256 answer, uint256 startedAt, uint256 updatedAt, uint80 answeredInRound);
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol


// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/token/ERC20/extensions/IERC20Metadata.sol


// OpenZeppelin Contracts (last updated v5.0.0) (token/ERC20/extensions/IERC20Metadata.sol)

pragma solidity ^0.8.20;


/**
 * @dev Interface for the optional metadata functions from the ERC20 standard.
 */
interface IERC20Metadata is IERC20 {
    /**
     * @dev Returns the name of the token.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the symbol of the token.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the decimals places of the token.
     */
    function decimals() external view returns (uint8);
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol


// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/utils/Context.sol


// OpenZeppelin Contracts (last updated v5.0.0) (utils/Context.sol)

pragma solidity ^0.8.20;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/access/Ownable.sol


// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;


/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: contracts/Knuck/Presale.sol


pragma solidity 0.8.21;





contract Presale is Ownable, ReentrancyGuard {
	event PhaseChanged(Phase newPhase);
	event Bought(address user, uint256 tokenAmount);
	event Claimed(address user, uint256 tokenAmount);

	error Presale__UnexpectedPhase();
	error Presale__CouldNotTransferUsdt();
	error Presale__CouldNotTransferToken();
	error Presale__NotEnoughTokens();
	error Presale__NotMinUsdAmount();
	error Presale__OverHardCap();

	enum Phase {
		INIT,
		BUY,
		CLAIM
	}

	IERC20Metadata public immutable token;
	IERC20Metadata public immutable usdt;
	AggregatorV3Interface public immutable ethUsdPriceFeed;

	address payable immutable presaleWallet;
	/// @notice 18 decimals
	uint256 public immutable usdPrice;
	/// @notice 18 decimals
	uint256 public immutable minUsdAmount;
	/// @notice 18 decimals
	uint256 public immutable usdHardCap;

	/// @notice 18 decimals
	uint256 public usdRaisedTotal;
	Phase public phase;
	uint256 public tokensToClaimTotal;
	mapping (address => uint256) public tokensToClaimByUser;

	modifier onlyPhase(Phase _phase) {
		if (phase != _phase) {
			revert Presale__UnexpectedPhase();
		}
		_;
	}

	constructor(
		IERC20Metadata _token,
		IERC20Metadata _usdt,
		AggregatorV3Interface _ethUsdPriceFeed,
		address payable _presaleWallet,
		uint256 _usdPrice,
		uint256 _minUsdAmount,
		uint256 _usdHardCap,
		address owner
	) {
		token = _token;
		usdt = _usdt;
		ethUsdPriceFeed = _ethUsdPriceFeed;
		presaleWallet = _presaleWallet;
		usdPrice = _usdPrice;
		minUsdAmount = _minUsdAmount;
		usdHardCap = _usdHardCap;
		_transferOwnership(owner);
	}

	function buyWithEth() external payable nonReentrant onlyPhase(Phase.BUY) {
		(uint256 usdAmount, uint256 tokenAmount) = this.getAmountsByEth(msg.value);

		if (usdAmount < minUsdAmount) {
			revert Presale__NotMinUsdAmount();
		}

		uint256 _usdRaisedTotal = usdRaisedTotal + usdAmount;
		if (_usdRaisedTotal > usdHardCap) {
			revert Presale__OverHardCap();
		}
		usdRaisedTotal = _usdRaisedTotal;

		presaleWallet.transfer(msg.value);

		_buy(tokenAmount);
	}

	function buyWithUsdt(uint256 usdtAmount) external nonReentrant onlyPhase(Phase.BUY) {
		(uint256 usdAmount, uint256 tokenAmount) = this.getAmountsByUsdt(usdtAmount);

		if (usdAmount < minUsdAmount) {
			revert Presale__NotMinUsdAmount();
		}

		uint256 _usdRaisedTotal = usdRaisedTotal + usdAmount;
		if (_usdRaisedTotal > usdHardCap) {
			revert Presale__OverHardCap();
		}
		usdRaisedTotal = _usdRaisedTotal;

		if (usdt.transferFrom(msg.sender, presaleWallet, usdtAmount) != true) {
			revert Presale__CouldNotTransferUsdt();
		}

		_buy(tokenAmount);
	}

	function getAmountsByEth(uint256 ethAmount) external view returns (uint256 usdAmount, uint256 tokenAmount) {
		(,int256 ethUsdPrice,,,) = ethUsdPriceFeed.latestRoundData();
		usdAmount = (ethAmount * uint256(ethUsdPrice)) / (10 ** ethUsdPriceFeed.decimals());
		tokenAmount = _getTokenAmountByUsd(usdAmount);
	}

	function getAmountsByUsdt(uint256 usdtAmount) external view returns (uint256 usdAmount, uint256 tokenAmount) {
		usdAmount = usdtAmount * (10 ** (18 - usdt.decimals()));
		tokenAmount = _getTokenAmountByUsd(usdAmount);
	}

	function claim() external nonReentrant onlyPhase(Phase.CLAIM) {
		uint256 amount = tokensToClaimByUser[msg.sender];
		tokensToClaimByUser[msg.sender] = 0;
		tokensToClaimTotal -= amount;

		if (token.transfer(msg.sender, amount) != true) {
			revert Presale__CouldNotTransferToken();
		}

		emit Claimed(msg.sender, amount);
	}

	function startPresale() external onlyOwner onlyPhase(Phase.INIT) {
		_changePhase(Phase.BUY);
	}

	function endPresale() external onlyOwner onlyPhase(Phase.BUY) {
		if (token.balanceOf(address(this)) < tokensToClaimTotal) {
			revert Presale__NotEnoughTokens();
		}

		_changePhase(Phase.CLAIM);
	}

	function _changePhase(Phase _phase) internal {
		phase = _phase;
		emit PhaseChanged(_phase);
	}

	function _getTokenAmountByUsd(uint256 usdAmount) internal view returns (uint256) {
		return (usdAmount / usdPrice) * 1e18;
	}

	function _buy(uint256 tokenAmount) internal {
		tokensToClaimByUser[msg.sender] += tokenAmount;
		tokensToClaimTotal += tokenAmount;

		emit Bought(msg.sender, tokenAmount);
	}
}